Solving for c_{in}, $j_{\text{in}}(= D\,\nabla c_{\text{in}})$ inside an active droplet with first order chemical reactions.
Solving for c_{out}, $j_{\text{out}}(= D\,\nabla c_{\text{out}})$ outside the droplet for an active/passive droplet with first order chemical reactions.

——

Note: Cells highlighted in green indicate relevant parts implemented in the code.

——

Inside the droplet:

We solve the steady state reaction-diffusion equation: $D\,\nabla^2 c_{\text{in}} + \text{KF}\,(1 - c_{\text{in}}) - \text{KB}\,(c_{\text{in}}) = 0$ analytically for c_{in} for 1, 2 and 3 dimensions.
This reaction flux scheme is specific to the first order reactions which convert droplet to background field inside the droplet with rates KF and KB.
(Note that for the passive case, $c_{\text{in}} = c_{\text{in}}^{\text{eq}}$ and hence the local fluxes evaluated at the droplet surface $j_{\text{in}} = 0$).
We integrate $(-\,D\nabla c_{\text{in}})$ at r = R over the surface of the droplet to get the integrated flux J_{in}.
Finally, the mean flux inside the droplet is calculated as J_{in}/(Volume of the droplet), which is used as an input to 'reaction_inside' in the code.

——

Outside the droplet:

We solve the steady state reaction-diffusion equation: $D\,\nabla^2 c_{out} +$ KF $(1 - c_{out})$ - KB $(c_{out}) =$ 0 analytically for c_{out} for 1, 2 and 3 dimensions.
This reaction flux scheme is specific to the first order reactions which convert droplet to background field inside the droplet with rates KF and KB.
We integrate (- D∇c_{out}) at r = R over the surface of the droplet to get the integrated flux J_{out}.

3D droplets:

Solve the stationary state equation inside the droplets:

```
In[116]:= ClearAll["Global`*"]

In[117]:= $Assumptions = Diffusion > 0 && KF > 0 && KB > 0 &&
             cEqin > 0 && cEqout > 0 && cInf > 0 && R > 0 && ξ > 0 && γ > 0 && L > R;

In[118]:= sol3DInside[r_] = FullSimplify[FullSimplify[
              c[r] /. First@DSolve[{0 == Diffusion * Laplacian[c[r], {r, ϕ, θ}, "Spherical"] +
                    KF * (1 - c[r]) - KB * (c[r]), Derivative[1][c][0] == 0,
                  c[R] == cEqin}, c, r]] /. Diffusion → (KF + KB) * ξ^2]
```

Out[118]= $\dfrac{\text{KF r} + (-\text{KF} + \text{cEqin}\,(\text{KB}+\text{KF}))\,\text{R Csch}\left[\frac{\text{R}}{\xi}\right]\,\text{Sinh}\left[\frac{\text{r}}{\xi}\right]}{(\text{KB}+\text{KF})\,\text{r}}$

Check the boundary conditions

```
In[119]:= FullSimplify@sol3DInside[R]

Out[119]= cEqin

In[120]:= Limit[D[sol3DInside[r], r], r → 0]

Out[120]= 0
```

Check the solution by plugging it in the Reaction-Diffusion equation

```
In[121]:= FullSimplify[
           FullSimplify[Diffusion * Laplacian[sol3DInside[r], {r, ϕ, θ}, "Spherical"] +
              KF * (1 - sol3DInside[r]) - KB * (sol3DInside[r])] /. Diffusion → (KF + KB) * ξ^2]

Out[121]= 0
```

Calculate surface integrated fluxes at the droplet surface

In[122]:=
```
TotalFluxInside3D =
  FullSimplify[FullSimplify[-4 π * R^2 * Diffusion * D[sol3DInside[r], r] /. r → R] /.
    Diffusion → (KF + KB) * ξ^2]
```

Out[122]= $4\,(-KF + cEqin\,(KB + KF))\,\pi\,R\,\xi\,\left(\xi - R\,\mathrm{Coth}\left[\frac{R}{\xi}\right]\right)$

In[123]:=
```
TotalFluxInside3DperVolume = FullSimplify[TotalFluxInside3D / ((4 / 3) π * R * R * R)]
```

Out[123]= $\frac{3\,(-KF + cEqin\,(KB + KF))\,\xi\,\left(\xi - R\,\mathrm{Coth}\left[\frac{R}{\xi}\right]\right)}{R^2}$

In[124]:=
```
LocalFluxInside3D = FullSimplify[-Diffusion * D[sol3DInside[r], r] /. r → R]
```

Out[124]= $\frac{\mathrm{Diffusion}\,(-KF + cEqin\,(KB + KF))\,\left(\xi - R\,\mathrm{Coth}\left[\frac{R}{\xi}\right]\right)}{(KB + KF)\,R\,\xi}$

Solve the stationary state equation outside the droplets:

In[125]:=
```
sol3DOutside[r_] = FullSimplify[Limit[FullSimplify[
     c[r] /. First@DSolve[{0 == Diffusion * Laplacian[c[r], {r, ϕ, θ}, "Spherical"] +
            KF * (1 - c[r]) - KB * (c[r]), c[R] == cEqout, c[L] == cInf}, c, r] /.
      Diffusion → (KF + KB) * ξ^2], L → Infinity]]
```

Limit: Warning: Assumptions that involve the limit variable are ignored.

Out[125]= $\frac{KF + \frac{e^{\frac{-r+R}{\xi}}\,(-KF+cEqout\,(KB+KF))\,R}{r}}{KB + KF}$

Check the boundary conditions

In[126]:=
```
FullSimplify@sol3DOutside[R]
```

Out[126]= cEqout

In[127]:=
```
FullSimplify@sol3DOutside[Infinity]
```

Out[127]= $\frac{KF}{KB + KF}$

Check the solution by plugging it in the Reaction-Diffusion equation

In[128]:=
```
FullSimplify[
  FullSimplify[Diffusion * Laplacian[sol3DOutside[r], {r, ϕ, θ}, "Spherical"] +
     KF * (1 - sol3DOutside[r]) - KB * (sol3DOutside[r])] /. Diffusion → (KF + KB) * ξ^2]
```

Out[128]= 0

Calculate surface area integrated fluxes at the droplet surface

In[129]:=
```
TotalFluxOutside3D =
  FullSimplify[FullSimplify[-4 π * R^2 * Diffusion * D[sol3DOutside[r], r] /. r → R]]
```

Out[129]= $$\frac{4\,\text{Diffusion}\,(-\text{KF}+\text{cEqout}\,(\text{KB}+\text{KF}))\,\pi\,\text{R}\,(\text{R}+\zeta)}{(\text{KB}+\text{KF})\,\zeta}$$

Calculate stable droplet size for Active emulsions

In[130]:=
```
LocalFluxOutside3D = FullSimplify[-Diffusion * D[sol3DOutside[r], r] /. r → R]
```

Out[130]= $$\frac{\text{Diffusion}\,(-\text{KF}+\text{cEqout}\,(\text{KB}+\text{KF}))\,(\text{R}+\zeta)}{(\text{KB}+\text{KF})\,\text{R}\,\zeta}$$

In[131]:=
```
KF = (1*^-5);
KB = (1*^-4);
γ = 0.0833;
cInf = KF / (KF + KB);
```

In[135]:=
```
InterfacialSpeed[R_] := FullSimplify[
   FullSimplify[(LocalFluxInside3D - LocalFluxOutside3D) / (cEqin - cEqout)] /.
       cEqout → (2 * γ / R) /. cEqin → (1 + 2 * γ / R) /.
     Diffusion → 1 /. ζ → Sqrt[1 / (KF + KB)]]
```

In[136]:=
```
InterfacialSpeed[R]
```

Out[136]= $$\frac{0.998253+0.000953463\,\text{R}+(-0.00174732-0.00953463\,\text{R})\,\text{Coth}\left[\frac{1}{100}\sqrt{\frac{11}{10}}\,\text{R}\right]}{\text{R}}$$

In[137]:=
```
CriticalRadiusNumerical = FindRoot[InterfacialSpeed[R], {R, 1}]
```

Out[137]= {R → 1.83504}

In[138]:=
```
StableRadiusNumerical = FindRoot[InterfacialSpeed[R], {R, 70}]
```

Out[138]= {R → 68.6091}

In[139]:=
```
CriticalRadiusTheory = 2 * γ / cInf
```

Out[139]= 1.8326

In[140]:=
```
StableRadiusTheory = Sqrt[3 * Diffusion * (KF / (KB + KF)) / KB] /. Diffusion → 1
```

Out[140]= $100\sqrt{\frac{3}{11}}$

In[141]:=
```
MeanReactionInside3D =
 Normal[Series[TotalFluxInside3DperVolume /. cEqin → (1 + 2 * γ / R) /.
     Diffusion → 1 /. ξ → Sqrt[1 / (KF + KB)], {R, 0, 4}]]
```

Out[141]=
$$-\frac{1}{10\,000}-\frac{0.000018326}{R}+1.34391\times 10^{-10}\,R+\frac{11\,R^2}{15\,000\,000\,000}-1.4079\times 10^{-15}\,R^3-\frac{121\,R^4}{15\,750\,000\,000\,000\,000}$$

In[142]:=
```
-KB
```

Out[142]=
$$-\frac{1}{10\,000}$$

In[143]:=

2D droplets:

Solve the stationary state equation inside the droplets:

In[144]:=
```
ClearAll["Global`*"]
```

In[145]:=
```
$Assumptions = Diffusion > 0 && KF > 0 && KB > 0 &&
   cEqin > 0 && cEqout > 0 && cInf > 0 && R > 0 && ξ > 0 && γ > 0 && L > R;
```

In[146]:=
```
sol2DInside[r_] =
 FullSimplify[FullSimplify[c[r] /. First@DSolve[{0 == Diffusion * Laplacian[c[r],
          {r, ϕ}, "Polar"] + KF * (1 - c[r]) - KB * (c[r]), Derivative[1][c][0] == 0,
        c[R] == cEqin}, c, r]] /. Diffusion → (KF + KB) * ξ^2]
```

Out[146]=
$$\frac{KF+\frac{(-KF+cEqin\,(KB+KF))\,\mathrm{BesselI}\left[0,\frac{r}{\xi}\right]}{\mathrm{BesselI}\left[0,\frac{R}{\xi}\right]}}{KB+KF}$$

Check the boundary conditions

In[147]:=
```
FullSimplify@sol2DInside[R]
```

Out[147]= cEqin

In[148]:=
```
Limit[D[sol2DInside[r], r], r → 0]
```

Out[148]= 0

Check the solution by plugging it in the Reaction-Diffusion equation

In[149]:=
```
FullSimplify[FullSimplify[Diffusion * Laplacian[sol2DInside[r], {r, ϕ}, "Polar"] +
    KF * (1 - sol2DInside[r]) - KB * (sol2DInside[r])] /. Diffusion → (KF + KB) * ξ^2]
```

Out[149]= 0

Calculate surface integrated fluxes at the droplet surface

In[150]:=
```
TotalFluxInside2D =
 FullSimplify[FullSimplify[-2 π * R * Diffusion * D[sol2DInside[r], r] /. r → R] /.
   Diffusion → (KF + KB) * ξ^2]
```

Out[150]= $$-\frac{2\,(-\mathrm{KF}+\mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\pi\,R\,\xi\,\mathrm{BesselI}\left[1,\frac{R}{\xi}\right]}{\mathrm{BesselI}\left[0,\frac{R}{\xi}\right]}$$

In[151]:=
```
TotalFluxInside2DperVolume = FullSimplify[TotalFluxInside2D / (π * R * R)]
```

Out[151]= $$-\frac{2\,(-\mathrm{KF}+\mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\xi\,\mathrm{BesselI}\left[1,\frac{R}{\xi}\right]}{R\,\mathrm{BesselI}\left[0,\frac{R}{\xi}\right]}$$

In[152]:=
```
LocalFluxInside2D = FullSimplify[-Diffusion * D[sol2DInside[r], r] /. r → R]
```

Out[152]= $$\frac{\mathrm{Diffusion}\,(\mathrm{KF}-\mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\mathrm{BesselI}\left[1,\frac{R}{\xi}\right]}{(\mathrm{KB}+\mathrm{KF})\,\xi\,\mathrm{BesselI}\left[0,\frac{R}{\xi}\right]}$$

Solve the stationary state equation outside the droplets:

In[153]:=
```
sol2DOutsideComplex[r_] = FullSimplify[c[r] /. First@DSolve[
     {0 == Diffusion * Laplacian[c[r], {r, ϕ}, "Polar"] + KF * (1 - c[r]) - KB * (c[r]),
      c[R] == cEqout, c[L] == cInf}, c, r] /. Diffusion → (KF + KB) * ξ^2]
```

Out[153]= $$\Big(\pi\Big(\mathrm{BesselI}\left[0,\tfrac{R}{\xi}\right]\Big(\mathrm{KF}\,\mathrm{BesselY}\left[0,-\tfrac{i L}{\xi}\right]+(\mathrm{cInf}\,\mathrm{KB}+(-1+\mathrm{cInf})\,\mathrm{KF})\,\mathrm{BesselY}\left[0,-\tfrac{i r}{\xi}\right]\Big)+\mathrm{BesselI}\left[0,\tfrac{L}{\xi}\right]\Big((\mathrm{KF}-\mathrm{cEqout}\,(\mathrm{KB}+\mathrm{KF}))\,\mathrm{BesselY}\left[0,-\tfrac{i r}{\xi}\right]-\mathrm{KF}\,\mathrm{BesselY}\left[0,-\tfrac{i R}{\xi}\right]\Big)+\mathrm{BesselI}\left[0,\tfrac{r}{\xi}\right]\Big((\mathrm{cEqout}\,\mathrm{KB}+(-1+\mathrm{cEqout})\,\mathrm{KF})\,\mathrm{BesselY}\left[0,-\tfrac{i L}{\xi}\right]+(\mathrm{KF}-\mathrm{cInf}\,(\mathrm{KB}+\mathrm{KF}))\,\mathrm{BesselY}\left[0,-\tfrac{i R}{\xi}\right]\Big)\Big)\Big)\Big/\Big(2\,(\mathrm{KB}+\mathrm{KF})\Big(-\mathrm{BesselI}\left[0,\tfrac{R}{\xi}\right]\mathrm{BesselK}\left[0,\tfrac{L}{\xi}\right]+\mathrm{BesselI}\left[0,\tfrac{L}{\xi}\right]\mathrm{BesselK}\left[0,\tfrac{R}{\xi}\right]\Big)\Big)$$

In[154]:=
```
sol2DOutside[r_] =
  FullSimplify[sol2DOutsideComplex[r] /. BesselY[0, -I * x_] :> -2 / π * BesselK[0, x]]
```

Out[154]=
$$\Big(\text{BesselI}\big[0,\tfrac{R}{\zeta}\big]\Big(\text{KF BesselK}\big[0,\tfrac{L}{\zeta}\big]+(\text{cInf KB}+(-1+\text{cInf})\,\text{KF})\,\text{BesselK}\big[0,\tfrac{r}{\zeta}\big]\Big)+\text{BesselI}\big[0,\tfrac{L}{\zeta}\big]\Big((\text{KF}-\text{cEqout}\,(\text{KB}+\text{KF}))\,\text{BesselK}\big[0,\tfrac{r}{\zeta}\big]-\text{KF BesselK}\big[0,\tfrac{R}{\zeta}\big]\Big)+\text{BesselI}\big[0,\tfrac{r}{\zeta}\big]\Big((\text{cEqout KB}+(-1+\text{cEqout})\,\text{KF})\,\text{BesselK}\big[0,\tfrac{L}{\zeta}\big]+(\text{KF}-\text{cInf}\,(\text{KB}+\text{KF}))\,\text{BesselK}\big[0,\tfrac{R}{\zeta}\big]\Big)\Big)\Big/\Big((\text{KB}+\text{KF})\Big(\text{BesselI}\big[0,\tfrac{R}{\zeta}\big]\,\text{BesselK}\big[0,\tfrac{L}{\zeta}\big]-\text{BesselI}\big[0,\tfrac{L}{\zeta}\big]\,\text{BesselK}\big[0,\tfrac{R}{\zeta}\big]\Big)\Big)$$

Check the boundary conditions

In[155]:=
```
FullSimplify@sol2DOutside[R]
```

Out[155]= cEqout

In[156]:=
```
FullSimplify@sol2DOutside[L]
```

Out[156]= cInf

Check the solution by plugging it in the Reaction-Diffusion equation

In[157]:=
```
FullSimplify[
  FullSimplify[Diffusion * Laplacian[sol2DOutside[r], {r, ϕ}, "Polar"] +
     KF * (1 - sol2DOutside[r]) - KB * (sol2DOutside[r])] /. Diffusion → (KF + KB) * ζ^2]
```

Out[157]= 0

Calculate surface area integrated fluxes at the droplet surface

In[158]:=
```
TotalFluxOutside2D =
  FullSimplify[FullSimplify[-2 π * R * Diffusion * D[sol2DOutside[r], r] /. r → R]]
```

Out[158]=
$$\Big(2\,\text{Diffusion}\,\pi\Big((-\text{KF}+\text{cInf}\,(\text{KB}+\text{KF}))\,\zeta+(\text{KF}-\text{cEqout}\,(\text{KB}+\text{KF}))\,R\Big(\text{BesselI}\big[1,\tfrac{R}{\zeta}\big]\,\text{BesselK}\big[0,\tfrac{L}{\zeta}\big]+\text{BesselI}\big[0,\tfrac{L}{\zeta}\big]\,\text{BesselK}\big[1,\tfrac{R}{\zeta}\big]\Big)\Big)\Big)\Big/\Big((\text{KB}+\text{KF})\,\zeta\Big(\text{BesselI}\big[0,\tfrac{R}{\zeta}\big]\,\text{BesselK}\big[0,\tfrac{L}{\zeta}\big]-\text{BesselI}\big[0,\tfrac{L}{\zeta}\big]\,\text{BesselK}\big[0,\tfrac{R}{\zeta}\big]\Big)\Big)$$

Calculate stable droplet size for Active emulsions

In[159]:=
```
LocalFluxOutside2D = FullSimplify[-Diffusion * D[sol2DOutside[r], r] /. r → R]
```

Out[159]=

$$\left(\text{Diffusion}\left(\frac{(-\text{KF}+\text{cInf}\,(\text{KB}+\text{KF}))\,\zeta}{R}+(\text{KF}-\text{cEqout}\,(\text{KB}+\text{KF}))\left(\text{BesselI}\left[1,\frac{R}{\zeta}\right]\text{BesselK}\left[0,\frac{L}{\zeta}\right]+\text{BesselI}\left[0,\frac{L}{\zeta}\right]\text{BesselK}\left[1,\frac{R}{\zeta}\right]\right)\right)\right)\Big/\left((\text{KB}+\text{KF})\,\zeta\left(\text{BesselI}\left[0,\frac{R}{\zeta}\right]\text{BesselK}\left[0,\frac{L}{\zeta}\right]-\text{BesselI}\left[0,\frac{L}{\zeta}\right]\text{BesselK}\left[0,\frac{R}{\zeta}\right]\right)\right)$$

In[160]:=
```
KF = (1*^-5);
KB = (1*^-4);
γ = 0.0833;
cInf = KF / (KF + KB);
L = 100 * R;
```

In[165]:=
```
InterfacialSpeed[R_] :=
 FullSimplify[(LocalFluxInside2D - LocalFluxOutside2D) / (cEqin - cEqout)] /.
     cEqout → (2 * γ / R) /. cEqin → (1 + 2 * γ / R) /.
   Diffusion → 1 /. ζ → Sqrt[1 / (KF + KB)]
```

In[166]:=
```
InterfacialSpeed[R]
```

Out[166]=

$$\left(0.000953463\times\left(-11.\,\text{BesselI}\left[0,\frac{1}{100}\sqrt{\frac{11}{10}}\,R\right]\text{BesselI}\left[1,\frac{1}{100}\sqrt{\frac{11}{10}}\,R\right]\text{BesselK}\left[0,\sqrt{\frac{11}{10}}\,R\right]+\text{BesselI}\left[0,\sqrt{\frac{11}{10}}\,R\right]\left(\left(-1+11\times\left(1+\frac{0.1666}{R}\right)\right)\text{BesselI}\left[1,\frac{1}{100}\sqrt{\frac{11}{10}}\,R\right]\text{BesselK}\left[0,\frac{1}{100}\sqrt{\frac{11}{10}}\,R\right]+\left(-1+\frac{1.8326}{R}\right)\text{BesselI}\left[0,\frac{1}{100}\sqrt{\frac{11}{10}}\,R\right]\text{BesselK}\left[1,\frac{1}{100}\sqrt{\frac{11}{10}}\,R\right]\right)\right)\right)\Big/\left(\text{BesselI}\left[0,\frac{1}{100}\sqrt{\frac{11}{10}}\,R\right]\left(-\text{BesselI}\left[0,\sqrt{\frac{11}{10}}\,R\right]\text{BesselK}\left[0,\frac{1}{100}\sqrt{\frac{11}{10}}\,R\right]+\text{BesselI}\left[0,\frac{1}{100}\sqrt{\frac{11}{10}}\,R\right]\text{BesselK}\left[0,\sqrt{\frac{11}{10}}\,R\right]\right)\right)$$

In[167]:=
```
CriticalRadiusNumerical = FindRoot[InterfacialSpeed[R], {R, 1}]
```

Out[167]=
```
{R → 1.84788}
```

In[168]:=
```
StableRadiusNumerical = FindRoot[InterfacialSpeed[R], {R, 40}]
```

Out[168]=
```
{R → 36.6037}
```

In[169]:=
```
CriticalRadiusTheory = 2 * γ / cInf
```

Out[169]= 1.8326

In[170]:=
```
MeanReactionInside2D =
 Normal[Series[TotalFluxInside2DperVolume /. cEqin → (1 + 2 * γ / R) /.
    Diffusion → 1 /. ξ → Sqrt[1 / (KF + KB)], {R, 0, 4}]]
```

Out[170]=

$$-\frac{1}{10000}-\frac{0.000018326}{R}+2.51983\times10^{-10}\,R+\frac{11\,R^{2}}{8000000000}-4.61968\times10^{-15}\,R^{3}-\frac{121\,R^{4}}{4800000000000000}$$

In[171]:=
```
-KB
```

Out[171]=

$$-\frac{1}{10000}$$

In[172]:=

1D droplets:

Solve the stationary state equation inside the droplets:

In[173]:=
```
ClearAll["Global`*"]
```

In[174]:=
```
$Assumptions = Diffusion > 0 && KF > 0 && KB > 0 &&
   cEqin > 0 && cEqout > 0 && cInf > 0 && R > 0 && ξ > 0 && γ > 0 && L > R;
```

In[175]:=
```
sol1DInside[r_] = FullSimplify[FullSimplify[
   c[r] /. First@DSolve[{0 == Diffusion * Laplacian[c[r], {r}, "Cartesian"] +
         KF * (1 - c[r]) - KB * (c[r]), Derivative[1][c][0] == 0,
        c[R] == cEqin}, c, r]] /. Diffusion → (KF + KB) * ξ^2]
```

Out[175]=

$$\frac{KF+(-KF+cEqin\,(KB+KF))\,\text{Cosh}\left[\frac{r}{\xi}\right]\,\text{Sech}\left[\frac{R}{\xi}\right]}{KB+KF}$$

Check the boundary conditions

In[176]:=
```
FullSimplify@sol1DInside[R]
```

Out[176]=
```
cEqin
```

In[177]:=
```
Limit[D[sol1DInside[r], r], r → 0]
```

Out[177]= 0

Check the solution by plugging it in the Reaction-Diffusion equation

In[178]:=
```
FullSimplify[
 FullSimplify[Diffusion * Laplacian[sol1DInside[r], {r}, "Cartesian"] +
    KF * (1 - sol1DInside[r]) - KB * (sol1DInside[r])] /. Diffusion → (KF + KB) * ξ^2]
```

Out[178]= 0

Calculate surface integrated fluxes at the droplet surface

In[179]:=
```
TotalFluxInside1D =
 FullSimplify[FullSimplify[-2 * Diffusion * D[sol1DInside[r], r] /. r → R] /.
   Diffusion → (KF + KB) * ξ^2]
```

Out[179]= $-2\,(-KF + cEqin\,(KB + KF))\,\xi\,\mathrm{Tanh}\left[\frac{R}{\xi}\right]$

In[180]:=
```
TotalFluxInside1DperVolume = FullSimplify[TotalFluxInside1D / (2 * R)]
```

Out[180]= $\frac{(KF - cEqin\,(KB + KF))\,\xi\,\mathrm{Tanh}\left[\frac{R}{\xi}\right]}{R}$

In[181]:=
```
LocalFluxInside1D = FullSimplify[-Diffusion * D[sol1DInside[r], r] /. r → R]
```

Out[181]= $\frac{Diffusion\,(KF - cEqin\,(KB + KF))\,\mathrm{Tanh}\left[\frac{R}{\xi}\right]}{(KB + KF)\,\xi}$

Solve the stationary state equation outside the droplets:

In[182]:=
```
sol1DOutside[r_] = FullSimplify[
  c[r] /. First@DSolve[{0 == Diffusion * Laplacian[c[r], {r}, "Cartesian"] +
         KF * (1 - c[r]) - KB * (c[r]), c[R] == cEqout,
       c[L] == cInf}, c, r] /. Diffusion → (KF + KB) * ξ^2]
```

Out[182]= $-\frac{1}{\left(-e^{\frac{2L}{\xi}} + e^{\frac{2R}{\xi}}\right)(KB + KF)}\,2\,e^{\frac{L+R}{\xi}}\left((-KF + cEqout\,(KB + KF))\,\mathrm{Sinh}\left[\frac{L - r}{\xi}\right] + KF\,\mathrm{Sinh}\left[\frac{L - R}{\xi}\right] + (cInf\,KB + (-1 + cInf)\,KF)\,\mathrm{Sinh}\left[\frac{r - R}{\xi}\right]\right)$

Check the boundary conditions

In[183]:=
```
FullSimplify@sol1DOutside[R]
```

Out[183]= cEqout

In[184]:=
```
FullSimplify@sol1DOutside[L]
```

Out[184]= cInf

Check the solution by plugging it in the Reaction-Diffusion equation

In[185]:=
```
FullSimplify[
 FullSimplify[Diffusion * Laplacian[sol1DOutside[r], {r}, "Cartesian"] +
    KF * (1 - sol1DOutside[r]) - KB * (sol1DOutside[r])] /. Diffusion → (KF + KB) * ξ^2]
```

Out[185]= 0

Calculate surface area integrated fluxes at the droplet surface

In[186]:=
```
TotalFluxOutside1D = FullSimplify[-2 * Diffusion * D[sol1DOutside[r], r] /. r → R]
```

Out[186]=
$$\frac{4\,\text{Diffusion}\,e^{\frac{L+R}{\xi}}\left(-\text{KF}+\text{cInf}\,(\text{KB}+\text{KF})+(\text{KF}-\text{cEqout}\,(\text{KB}+\text{KF}))\,\text{Cosh}\left[\frac{L-R}{\xi}\right]\right)}{\left(-e^{\frac{2L}{\xi}}+e^{\frac{2R}{\xi}}\right)(\text{KB}+\text{KF})\,\xi}$$

Calculate stable droplet size for Active emulsions

In[187]:=
```
LocalFluxOutside1D =
 FullSimplify[FullSimplify[-Diffusion * D[sol1DOutside[r], r] /. r → R]]
```

Out[187]=
$$\frac{2\,\text{Diffusion}\,e^{\frac{L+R}{\xi}}\left(-\text{KF}+\text{cInf}\,(\text{KB}+\text{KF})+(\text{KF}-\text{cEqout}\,(\text{KB}+\text{KF}))\,\text{Cosh}\left[\frac{L-R}{\xi}\right]\right)}{\left(-e^{\frac{2L}{\xi}}+e^{\frac{2R}{\xi}}\right)(\text{KB}+\text{KF})\,\xi}$$

In[188]:=
```
KF = (1*^-5);
KB = (1*^-4);
γ = 0.0833;
cInf = KF / (KF + KB);
L = 100 * R;
```

In[193]:=
```
InterfacialSpeed[R_] := FullSimplify[
  FullSimplify[(LocalFluxInside1D - LocalFluxOutside1D) / (cEqin - cEqout)] /.
     cEqout → (2 * γ / R) /. cEqin → (1 + 2 * γ / R) /.
   Diffusion → 1 /. ξ → Sqrt[1 / (KF + KB)]]
```

In[194]:=
```
CriticalRadiusNumerical = FindRoot[InterfacialSpeed[R], {R, 1.83}]
```

Out[194]= {R → 2.56409}

In[195]:=
```
StableRadiusNumerical = FindRoot[InterfacialSpeed[R], {R, 7}]
```

Out[195]= {R → 6.78963}

In[196]:=
```
CriticalRadiusTheory = 2 * γ / cInf
```

Out[196]= 1.8326

In[197]:=
```
MeanReactionInside1D =
 Normal[Series[TotalFluxInside1DperVolume /. cEqin → (1 + 2 * γ / R) /.
    Diffusion → 1 /. ξ → Sqrt[1 / (KF + KB)], {R, 0, 4}]]
```

Out[197]=
$$-\frac{1}{10\,000} - \frac{0.000018326}{R} + 6.71953 \times 10^{-10}\, R + \frac{11\, R^2}{3\,000\,000\,000} - 2.95659 \times 10^{-14}\, R^3 - \frac{121\, R^4}{750\,000\,000\,000\,000}$$

In[198]:=
```
-KB
```

Out[198]=
$$-\frac{1}{10\,000}$$